#tcp_send_ack tcp_send_delayed_ack-e1000_receive_skb
p $tmp=0
while $tmp<$cpu_number
  eval "tvariable $per_cpu_c_%d",$tmp
  eval "tvariable $per_cpu_tmp_0_%d",$tmp
  eval "tvariable $per_cpu_tmp_1_%d",$tmp
  eval "tvariable $per_cpu_cy1_%d",$tmp
  eval "tvariable $per_cpu_c1_%d",$tmp
  eval "tvariable $per_cpu_cy2_%d",$tmp
  eval "tvariable $per_cpu_c2_%d",$tmp
  eval "tvariable $per_cpu_go_%d",$tmp
  p $tmp=$tmp+1
end

list tcp_rcv_established
trace 5322
  commands
    teval $per_cpu_go_0=1
  end

trace tcp_send_ack if ($per_cpu_go_0)
  commands
    teval $per_cpu_c_0=$per_cpu_c_0+1
    teval $per_cpu_tmp_0_0=$clock
    collect *(unsigned char *)$rsp@512
  end

trace tcp_send_delayed_ack if ($per_cpu_go_0)
  commands
    teval $per_cpu_c_0=$per_cpu_c_0+1
    teval $per_cpu_tmp_1_0=$clock
    collect *(unsigned char *)$rsp@512
  end

list e1000_receive_skb
trace 774 if ($per_cpu_c_0>5 && $per_cpu_tmp_0_0 != 0)
  commands
    teval $per_cpu_cy1_0=$clock-$per_cpu_tmp_0_0+$per_cpu_cy1_0
    teval $per_cpu_c1_0=$per_cpu_c1_0+1
    teval $per_cpu_tmp_0_0=0
  end

list e1000_receive_skb
trace 774 if ($per_cpu_c_0>5 && $per_cpu_tmp_1_0 != 0)
  commands
    teval $per_cpu_cy2_0=$clock-$per_cpu_tmp_1_0+$per_cpu_cy2_0
    teval $per_cpu_c2_0=$per_cpu_c2_0+1
    teval $per_cpu_tmp_1_0=0
  end

define showall
p $per_cpu_c1_0
p $per_cpu_cy1_0/$per_cpu_c1_0
p $per_cpu_c2_0
p $per_cpu_cy2_0/$per_cpu_c2_0
end
